Необходимо в обработке за период создавать на каждую дату по консолидированному документу. Запросом получаю документы за период (Между &ДатаНачало и &ДатаКонец) . Есть ПКО 2 штуки за одну дату и 3 за другую - нужно создать налоговые накладные на каждую дату отдельно. Что то не огу сообразить как сделать.

```1c
    "ВЫБРАТЬ
    | ПриходныйКассовыйОрдерПриемРозничнойВыручки.НоменклатурнаяГруппа,
    | СУММА(ПриходныйКассовыйОрдерПриемРозничнойВыручки.Сумма) КАК Сумма,
    | ЗначенияСвойствОбъектов.Объект,
    | ЕСТЬNULL(ЗначенияСвойствОбъектов.Значение, "") КАК Код,
    | СУММА(ПриходныйКассовыйОрдерПриемРозничнойВыручки.СуммаАкцизногоНалога) КАК СуммаАкцизногоНалога,
    | СУММА(ПриходныйКассовыйОрдерПриемРозничнойВыручки.СуммаНДС) КАК СуммаНДС
    |ИЗ
    | Документ.ПриходныйКассовыйОрдер.ПриемРозничнойВыручки КАК ПриходныйКассовыйОрдерПриемРозничнойВыручки
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЗначенияСвойствОбъектов КАК ЗначенияСвойствОбъектов
    | ПО ПриходныйКассовыйОрдерПриемРозничнойВыручки.НоменклатурнаяГруппа.Ссылка = ЗначенияСвойствОбъектов.Объект
    | И (ЗначенияСвойствОбъектов.Свойство.Наименование = "Код УКТВЭД")
    |ГДЕ
    | ПриходныйКассовыйОрдерПриемРозничнойВыручки.Ссылка.Дата МЕЖДУ &ДатаНачала И &ДатаКонец
    | И ПриходныйКассовыйОрдерПриемРозничнойВыручки.Ссылка.ПометкаУдаления = ЛОЖЬ
    | И ПриходныйКассовыйОрдерПриемРозничнойВыручки.Ссылка.Проведен = ИСТИНА
    |
    |СГРУППИРОВАТЬ ПО
    | ПриходныйКассовыйОрдерПриемРозничнойВыручки.НоменклатурнаяГруппа,
    | ЗначенияСвойствОбъектов.Объект,
    | ЗначенияСвойствОбъектов.Значение,
    | ЕСТЬNULL(ЗначенияСвойствОбъектов.Свойство, "")"
```
---
Вот так надо:

```1c
    Выборка = РезультатЗапроса.Выбрать(ОбходРезультатаЗапроса.ПоГруппировкам);
```

Первая выборка будет по верхнему итогу.
Внутри первой выборки можно сделать выборку из выборки, уже прямым методом, получишь детальные записи внутри периода, вот ак:

```1c
    ВыборкаПоНомГруппам = Выборка.Выбрать();
```
---
Добавь итог
НАЧАЛОПЕРИОДА(ПриходныйКассовыйОрдерПриемРозничнойВыручки.Дата,ДЕНЬ)
И по нему обход делай
---
можно ещё подсказку , не понимаю что то до конца у меня вот такой код куда вставить , как обойти по итогу НАЧАЛОПЕРИОДА

```1c
    РезультатЗапроса = Запрос.Выполнить();
    Если РезультатЗапроса.Пустой() тогда
    Возврат;
    иначе
    Выборка = РезультатЗапроса.Выбрать();

    НалоговыйДокумент = Документы.НалоговаяНакладная.СоздатьДокумент();
    НалоговыйДокумент.ВидОперации = Перечисления.ВидыОперацийНалоговаяНакладная.ИтоговаяРозницаОблагаемыеОперации;
    //Заполним шапку документа
    ЗаполнитьШапку(НалоговыйДокумент, ,ДатаНачала);
    НалоговыйДокумент.Товары.Очистить();
    НалоговыйДокумент.Услуги.Очистить();

    Пока Выборка.Следующий() Цикл
    НоваяСтрока = НалоговыйДокумент.Товары.Добавить();
    НоваяСтрока.Количество = 1;
```
---
Вот так надо:

```1c
    Выборка = РезультатЗапроса.Выбрать(ОбходРезультатаЗапроса.ПоГруппировкам);
```

Первая выборка будет по верхнему итогу.
Внутри первой выборки можно сделать выборку из выборки, уже прямым методом, получишь детальные записи внутри периода, вот ак:

```1c
    ВыборкаПоНомГруппам = Выборка.Выбрать();
```
---
что то недопонимаю никак ,простите но можно ещё раз никак до меня не доходят эти группировки . что у меня не правильно?

```1c
    ВыборкаНачалоПериода = РезультатЗапроса.Выбрать(ОбходРезультатаЗапроса.ПоГруппировкам);

    Пока ВыборкаНачалоПериода.Следующий() Цикл
    НалоговыйДокумент = Документы.НалоговаяНакладная.СоздатьДокумент();
    НалоговыйДокумент.ВидОперации = Перечисления.ВидыОперацийНалоговаяНакладная.ИтоговаяРозницаОблагаемыеОперации;
    //Заполним шапку документа
    ЗаполнитьШапку(НалоговыйДокумент, ,ДатаНачала);
    НалоговыйДокумент.Товары.Очистить();
    НалоговыйДокумент.Услуги.Очистить();
    Выборка = ВыборкаНачалоПериода.Следующий();
    Пока Выборка.Следующий() Цикл
    НоваяСтрока = НалоговыйДокумент.Товары.Добавить();
```
---
```1c
    Выборка = ВыборкаНачалоПериода.Следующий();
```

Тут ошибка. Должно быть

```1c
    Выборка = ВыборкаНачалоПериода.Выбрать();
```

В запросе ИТОГИ ПО добавил?
---
добавил

```1c
    |ИТОГИ ПО
    | НАЧАЛОПЕРИОДА(ПриходныйКассовыйОрдерПриемРозничнойВыручки.Ссылка.Дата, ДЕНЬ) КАК НачалоПериода";
```

да там ошибка , исправил , теперь два документа то что нужно. только что то всего сумма по документа не получилась и с датой не могу понять как проставить
---
В конструкторе запроса где группировочное поле добавлял в нижнюю таблицу надо добавить суммируемое поле.
С датой не понятен вопрос. Куда поставить?
---
Сделал так сейчас попробую, но с суммой я ещё разберусь, а вот с датой не знаю что нужно брать, куда поставить вроде бы видно

```1c
    |СГРУППИРОВАТЬ ПО
    | ПриходныйКассовыйОрдерПриемРозничнойВыручки.НоменклатурнаяГруппа,
    | ЗначенияСвойствОбъектов.Объект,
    | ЗначенияСвойствОбъектов.Значение,
    | ЕСТЬNULL(ЗначенияСвойствОбъектов.Свойство, ""),
    | ПриходныйКассовыйОрдерПриемРозничнойВыручки.Ссылка.Дата
    |ИТОГИ
    | СУММА(Сумма)
    |ПО
    | НАЧАЛОПЕРИОДА(ПриходныйКассовыйОрдерПриемРозничнойВыручки.Ссылка.Дата, ДЕНЬ) КАК НачалоПериода"
```

дата у меня вот здесь сейчас ставлю параметр запроса он же реквизит , понимаю что неправильно но как взять дату по группировкам опять же не знаю , сейчас у меня вот так

```1c
    Пока ВыборкаНачалоПериода.Следующий() Цикл
    НалоговыйДокумент = Документы.НалоговаяНакладная.СоздатьДокумент();
    НалоговыйДокумент.ВидОперации = Перечисления.ВидыОперацийНалоговаяНакладная.ИтоговаяРозницаОблагаемыеОперации;
    //Заполним шапку документа
    ЗаполнитьШапку(НалоговыйДокумент, ,ДатаНачала);
    НалоговыйДокумент.Товары.Очистить();
    НалоговыйДокумент.Услуги.Очистить();
    Выборка = ВыборкаНачалоПериода.Выбрать();
    Пока Выборка.Следующий() Цикл
    НоваяСтрока = НалоговыйДокумент.Товары.Добавить();
    НоваяСтрока.Количество = 1;
```
---
Дату нового документа? Я правильно понял?

```1c
    ВыборкаНачалоПериода.НачалоПериода
```
 и будет нужная дата с временем 0:00
---
Да дата нового документа. я тоже думал что так ВыборкаНачалоПериода.НачалоПериода , если я правильно понимаю то НачалоПериода это в запросе

```1c
    |ПО
    | НАЧАЛОПЕРИОДА(ПриходныйКассовыйОрдерПриемРозничнойВыручки.Ссылка.Дата, ДЕНЬ) КАК НачалоПериода"
```

но пишет поле не обнаружено. наверное в этой выборке нету даты, или опять где то ошибся
---
Так ты поле то добавь. Начало запроса должно быть так:

```1c
    "ВЫБРАТЬ
    | НАЧАЛОПЕРИОДА(ПриходныйКассовыйОрдерПриемРозничнойВыручки.Ссылка.Дата, ДЕНЬ) КАК НачалоПериода,
    | ПриходныйКассовыйОрдерПриемРозничнойВыручки.НоменклатурнаяГруппа,
    | СУММА(ПриходныйКассовыйОрдерПриемРозничнойВыручки.Сумма) КАК Сумма,
    | ЗначенияСвойствОбъектов.Объект,
    | ЕСТЬNULL(ЗначенияСвойствОбъектов.Значение, "") КАК Код,
    | СУММА(ПриходныйКассовыйОрдерПриемРозничнойВыручки.СуммаАкцизногоНалога) КАК СуммаАкцизногоНалога,
    | СУММА(ПриходныйКассовыйОрдерПриемРозничнойВыручки.СуммаНДС) КАК СуммаНДС
    |ИЗ
    | Документ.ПриходныйКассовыйОрдер.ПриемРозничнойВыручки КАК ПриходныйКассовыйОрдерПриемРозничнойВыручки
```
---
Тогда в итоги можно написать так:

```1c
    |ИТОГИ
    | СУММА(Сумма)
    |ПО
    | НачалоПериода"
```
---
Есть заработало, спасибо огромное. Более менее понял про обход по группировкам.